

HT47C20 定时/计数器（Timer/Event counter）使用介绍

作 者： 盛扬半导体（上海）有限公司软件部


日 期： 2001/8/6

适用单片机： HT47C20、HT47C20L

本文主要介绍 HT47C20 定时/计数器（Timer/Event counter）的使用及注意事项。

介绍

HT47C20 有一个 16 位向上计数的定时/计数器，由 TMRAH、TMRAL 和 TMRBH、TMRBL 组成。其中 TMRAH、TMRAL 用来计数；TMRBH、TMRBL 是加载寄存器，用来存放定时/计数器的初值。定时/计数器赋初值时要先写入低字节，再写入高字节，即先写入 TMRAL、TMRBL，再写入 TMRAH、TMRBH；定时/计数器的时钟来源可以是系统时钟、指令时钟（系统时钟/4）、RTC 超时时钟或是外部时钟。其结构图如下：



TMRC 为定时/计数器的控制寄存器，用来定义定时/计数器的运行模式、计数允许/禁止以及触发方式，如下表所示：

名称	位	功能
—	0~2	未定义，读取时为“0”
TE	3	定义定时/计数器 TMR 作用沿（0=上升沿作用，1=下降沿作用）
TON	4	允许/禁止定时器计数（0=禁止，1=允许）
TN0 TN1 TN2	5 6 7	定义操作方式（TN2,TN1,TN0） 000=定时器模式（系统时钟） 001=定时器模式（系统时钟/4） 010=定时器模式（RTC 输出） 011=A/D 计数模式（由 ADCR 寄存器决定 RC 振荡） 100=计数器模式（外部时钟） 101=脉冲宽度测量模式（系统时钟/4） 110=未定义 111=未定义

TMRC 寄存器

TN0、TN1 和 TN2 用来定义计数模式。其中定时器模式用来计数内部时钟，其时钟来源为内部时钟；计数器模式用来计数外部事件，其时钟来源为外部 TMR 引脚的输入；A/D 计数模式是用

来计数外部 A/D 输入（关于 A/D 转换的使用在此不做详细介绍）；脉冲宽度测量模式能够对外部引脚 TMR 的高电平或低电平的持续时间进行计数，计数的时钟来源为指令时钟。

在定时器模式、A/D 计数模式或计数器模式下，一旦定时/计数器开始计数即从定时/计数器的现有值（TMRAH 和 TMRAL）开始计数至 FFFFH。若发生溢出，计数器即从定时/计数器加载寄存器（TMRBH 和 TMRBL）重新装入加载值，并同时置位中断请求标志（TF；INTC1 的第四位）。

在脉冲宽度测量模式下，当 TON 和 TE 位的值为 1 时，如果 TMR 收到由低电平到高电平（如果 TE 位的值为“0”，则为由高电平到低电平）的转变信号，计数器就会开始计数，直到 TMR 引脚回到原来的电平为止，并且会将 TON 位清零。测量的结果会依然存放在定时/计数器之中，也就是说一次只能计数一次脉冲的宽度。而当 TON 位重新置位为“1”，只要 TMR 收到跳变脉冲，测量就会再次执行下去。在脉冲测量模式中，定时/计数器并不会根据逻辑电压来计数，其根据的标准为信号的转变沿。一旦发生计数溢出，计数器会从定时/计数器加载寄存器重新装入初值，同时还会发出中断请求，这种情况和定时和计数这两个模式一样。

若欲启动计数器运行，只要将定时器启动位（TON；TMRC 的第四位）的值设为“1”即可。在脉冲宽度测量模式中，TON 位在测量周期完成后，会自动被清除。但在其它三种模式中，TON 位只可以用指令清除。定时/计数器的溢出可作为唤醒的信号或由掩膜选择设定为 PFD 输出。不管是何种操作模式，只要将“0”写入 ETI 位中即可将相对的中断服务禁止。


若在定时/计数器关闭的情况下，将数据写入定时/计数器的加载寄存器同时也会将该数据重新载入定时/计数器之中。但若定时/计数器已经开启，写入定时/计数器的数据只会保存在定时/计数器的加载寄存器中。这时定时/计数器并不会马上被改变而会继续计数下去，直到发生溢出为止，此时再由加载寄存器装入新的初始值。

在读取定时/计数器的数据时，会将时钟禁止，以避免发生错误。将时钟禁止可能会导致计数错误，在编程时需加以注意。

定时/计数器（Timer/Event counter）的使用

硬件部分：

PA3~PA7 引脚分别接发光二极管 LED，如图：



1. 定时器模式

当定时/计数器工作于定时器模式时，其时钟来源可以由指令选择为系统时钟、指令时钟（系统时钟/4）或实时时钟超时信号。

时钟来源	频率
系统时钟	480kHz
指令时钟	120kHz
实时时钟	32.768kHz

程序清单：

```
； 文件名：timer1.asm
； 说明：定时器模式的使用
include ht47c20.inc
data .section 'data'
int_count db ?         ； 中断次数寄存器
io_count  db ?         ； I/O 口状态寄存器
code .section at 0 'code'
org 00h
jmp start
org 04h
reti
org 08h
reti
org 0ch
reti
org 10h
jmp timer_int         ； 定时/计数器中断入口
； -----
org 20h
start:
clr intc0
clr intc1
clr adcr.1
set tmrc.5             ； 以指令时钟为计数时钟（120kHz）
mov a,0ah
mov int_count,a
clr io_count
mov a,LOW(65536-12000) ； 定时/计数器初值
mov tmral,a           ； 0.1 秒产生中断
mov tmrb1,a
mov a,HIGH(65536-12000)
mov tmrah,a
mov tmrbh,a
set tmrc.4             ； 开定时/计数器
set intc1.0
set intc0.0
```

```

        jmp     $
; -----
timer_int:
        sdz     int_count          ; 判断中断次数是否满 10 次
        reti                    ; 未满 10 次，则返回
        inc     io_count           ; 满 10 次，则 io_count 增 1
        swapa   io_count           ; 并将结果显示于 LED
        cpl     acc
        mov     pa,a
        mov     a,0ah              ; int_count 重新初始化
        mov     int_count,a
        reti

```

程序说明：

定时器每 0.1 秒产生一次中断，以 int_count 来记录中断次数，当中断次数达到 10 次（即 1 秒钟），则 io_count 加 1，并通过 LED 显示出来。有 4 个 LED，以二进制形式显示秒数，这样总共可以显示 16 秒。

2. 计数器模式

当定时/计数器工作于计数器模式时，其时钟来源为外部 TMR 引脚的输入信号。

程序清单：

```

; 文件名: timer2.asm
; 说明: 外部计数模式的使用
include ht47c20.inc
data .section 'data'
count db ?          ; 中断次数寄存器
code .section at 0 'code'
org     00h
jmp     start
org     04h
reti
org     08h
reti
org     0ch
reti
org     10h
jmp     timer_int    ; 定时/计数器中断入口
; -----
start:
        clr     intc0
        clr     intc1
        clr     adcr.1      ; 定时/计数器允许
        set     tmrc.7      ; 计数外部时钟
        clr     count
        mov     a,LOW(65536-1000) ; 定时/计数器初值

```

```

mov    tmral,a                ; 计数 1000 个外部事件产生中断
mov    tmrbl,a
mov    a,HIGH(65536-1000)
mov    tmrah,a
mov    tmrbh,a
set    tmrc.4                 ; 打开定时/计数器
set    intc1.0                ; 定时/计数器中断允许
set    intc0.0                ; 总中断允许
jmp    $

; -----
timer_int:
inc     count                 ; 中断次数增 1
swapa   count                 ; 将结果显示于 LED
cpl     acc
mov     pa,a
reti

```

程序说明:

定时/计数器计数外部事件，每计数满 1000 产生中断，并将中断次数通过 LED 显示出来。有 4 个 LED，以二进制形式显示中断次数，这样总共可以显示 16 次中断，即最多可显示 16000 次外部事件。

3. 脉冲宽度测量模式

脉冲宽度测量模式可以用来测量外部引脚输入信号高、低电平持续的时间。

程序清单:

```

; 文件名: timer3.asm
; 说明: 脉冲宽度测量模式的使用
include ht47c20.inc
data .section 'data'
count db ?                  ; 中断次数寄存器
code .section at 0 'code'
org    00h
jmp     start
org    04h
reti
org    08h
reti
org    0ch
reti
org    10h
jmp     timer_int           ; 定时/计数器中断入口
; -----
start:
clr     intc0
clr     intc1

```

```
clr    adcr.1
set    tmrc.5                ; 脉宽测量模式，计数时钟为指令时钟
set    tmrc.7
clr    count
mov     a, LOW(65536-60000)   ; 定时/计数器初值
mov     tmral, a              ; 每 0.5 秒产生中断
mov     tmrbl, a
mov     a, HIGH(65536-60000)
mov     tmrah, a
mov     tmrbh, a
set     tmrc.4
set     intc1.0
set     intc0.0
jmp     $
; -----
timer_int:
inc     count                ; 中断次数增 1
swapa   count                ; 将结果显示于 LED
cpl     acc
mov     pa, a
reti
```

程序说明：

在脉冲宽度测量模式中，定时/计数器每 0.5 秒产生一次中断，并将结果显示于 LED。有 4 个 LED，以二进制形式显示，这样总共可以显示 16 次中断，即最多可显示 8 秒的脉冲宽度。